

# Lecture twenty-three: Strategies for Nonproportional Hazards Data

## Cox model with stratification and piece-wise Cox model

As we saw in previous lectures, there are instances when PH assumption is violated for some covariates. Another example of nonproportional hazards is comparison of a surgical procedure with chemotherapy in section 11.1 (page 381). In such cases, we can solve the problem by introducing time-dependent covariates in Cox model including piece-wise Cox model or a different model (eg: AFT). Another option is to stratify on that variable and employ the proportional hazards model within each stratum for other covariates.

### 1. Stratified PH models

The hazard function is specified as following

$$h_{ij}(t) = \exp(\beta' \mathbf{x}_{ij}) h_{0j}(t),$$

where  $i = 1, \dots, n_j$ ,  $j = 1, \dots, g$ , and  $\mathbf{x}_{ij}$  is the vector of values of  $p$  explanatory variables,  $X_1, \dots, X_p$ , recorded on the  $i$ th individual in the  $j$ th stratum.

- (a) Assume different and completely unrelated, but arbitrary, baseline hazard  $h_{0j}$  in each stratum.
- (b) Assume the same  $\beta$  for all strata.
- (c) Stratification is completely effective in removing the problem of non-proportionality.
- (d) **Drawback:** Because stratum effects are modeled nonparametrically, there are no immediate tests of the null hypothesis of no association between a stratification factor and survival.
- (e) Stratification works naturally for categorical variables, for example, clinical centers. Quantitative variables can always be discretized, but it is not always obvious how to do so.

### 2. The likelihood of stratified PH model

The partial log likelihood is

$$LL(\beta) = [LL_1(\beta)] + [LL_2(\beta)] + \dots + [LL_g(\beta)],$$

where  $LL_j(\beta)$  is log partial likelihood using only the data for those individuals in the  $j$ th stratum.

- (a) Estimation and hypothesis testing methods are the same as that for proportional hazards model described in chapter 3.
- (b) A key assumption: the covariates are acting similarly on the baseline hazard function in each stratum.
- (c) The above assumption can be tested by using a likelihood ratio test or a Wald test. Assume the coefficients in each stratum is  $\mathbf{b}_j$ , under the null hypothesis (i.e.  $\beta$ 's are the same in each stratum),

$$-2[LL(\hat{\beta}) - \sum_{j=1}^g LL_j(\hat{\mathbf{b}}_j)]$$

has a large-sample,  $\chi^2((g-1)p)$ .

- 3. The stratified PH model can also be used to model matched pair experiments (see the example below).
- 4. The large sample stratified tests of hypotheses on regression coefficients are appropriate when either the sample size within strata is large or when the number of strata is large.
- 5. Example:  $P^2C^2$  Study:

The pediatric pulmonary and cardiovascular complications ( $P^2C^2$ ) of vertically transmitted human immunodeficiency virus (HIV) infection study is a multi-center prospective natural history study funded by national heart, lung and blood institute and coordinated by the Department of Biostatistics and Epidemiology of CCF. The data set here is subpopulation (group II a) of the original study and following variables are selected from many variables for illustration only:

time: time from birth to death  
dcode: censoring variable (0 = censored, 1 = died)  
ccd4: CD4 T-cell counts (only the first measurement, baseline?)  
cencat: medical centers participated in the study (cencat = 1, 2, 3)

- (a) Some people prefer stratification by center even if the PH assumption for center (cencat) is not violated.

- (b) Testing  $H_0 : \beta_{11} = \beta_{21} = \beta_{31}$   
 Thus,  $178.048 - 66.959 - 102.98 - 5.78 = 3.212$ , the degree of

Table 1: Results from Cox models for  $P^2C^2$  study

model	$-2\log L$
Without stratification	234.958
Stratified by center	178.048
Stratum: <i>cencat</i> = 1	66.959
Stratum: <i>cencat</i> = 2	102.98
Stratum: <i>cencat</i> = 3	5.78

freedom is 2, so the p-value is 0.201, suggesting no evidence against the null hypothesis.

- (c) SAS output:  
 The results from two models are

Model without stratification:

Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
CCD4	1	-0.000226	0.0001718	1.72338	0.1893	1.000

Model stratified by center:

Analysis of Maximum Likelihood Estimates						
Variable	DF	Parameter Estimate	Standard Error	Wald Chi-Square	Pr > Chi-Square	Risk Ratio
CCD4	1	-0.000211	0.0001819	1.34597	0.2460	1.000

- (d) SAS program:

```

options ls = 80;
libname fu '.';
data w;
    set fu.p2c2;
proc phreg;
    model time*dcode(0) = ccd4;
proc phreg;
    model time*dcode(0) = ccd4;
    strata cencat;
data w1;
    set fu.p2c2;
if cencat = 1;
proc phreg;
    model time*dcode(0) = ccd4;
data w2;
    set fu.p2c2;
if cencat = 2;
proc phreg;
    model time*dcode(0) = ccd4;
data w3;
    set fu.p2c2;
if cencat = 3;
proc phreg;
    model time*dcode(0) = ccd4;
run;

```

(e) p2c2.sas7bdat and above sas program are available at the course website.

6. Example (from Klein and Moeschberger, 2003):

Freireich et al. (1963) report the results of a clinical trial of a drug 6-mercaptopurine (6-MP) versus a placebo in 42 children with acute leukemia (Clinical trial: 6-MP versus placebo). The trial was conducted by matching pairs of patients at a given hospital by remission status (complete or partial) and randomizing within the pair to either a 6-MP or placebo maintenance therapy.

Data can be read in free format. The variables represented in the dataset are as follows:

Pair

Remission status at randomization (1=partial, 2=complete)

Time to relapse for placebo patients, months

Time to relapse for 6-MP patients, months

Relapse indicator (0=censored, 1=relapse) for 6-MP patients

NOTE: All placebo patients relapsed

Reference Freireich et al. Blood 21(1963): 699-716.

(a) The data: available at the course website

Obs	pair	rstatus	t1	t2	censor
1	1	1	1	10	1
2	2	2	22	7	1
3	3	2	3	32	0
4	4	2	12	23	1
5	5	2	8	22	1
6	6	1	17	6	1
7	7	2	2	16	1
8	8	2	11	34	0
9	9	2	8	32	0
10	10	2	12	25	0
11	11	2	2	11	0
12	12	1	5	20	0
13	13	2	4	19	0
14	14	2	15	6	1
15	15	2	8	17	0
16	16	1	23	35	0
17	17	1	5	6	1
18	18	2	11	13	1
19	19	2	4	9	0
20	20	2	1	6	0
21	21	2	8	10	0

(b) manipulate the data:

The SAS program:

```
options ls = 80;
libname fu '.';
data tmp0;
    infile './aml.dat';
    input pair rstatus t1 t2  censor;
proc print;
data tmp1 (drop = t1 t2);
    set tmp0;
    array st[2] t1-t2;
    do treat=1 to 2;
        survt=st[treat];
        output;
    end;
data fu.aml;
    set tmp1;
/* all patients in placebo group had event */
    if treat = 1 then censor = 1;
run;
```

(c) Fit the data: Output from PROC PHREG:

#### Model Fit Statistics

Criterion	Without Covariates	With Covariates
-2 LOG L	29.112	17.225
AIC	29.112	19.225
SBC	29.112	20.626

#### The PHREG Procedure

Testing Global Null Hypothesis: BETA=0

Test	Chi-Square	DF	Pr > ChiSq
------	------------	----	------------

Likelihood Ratio	11.8873	1	0.0006
Score	10.7143	1	0.0011
Wald	8.2553	1	0.0041

Analysis of Maximum Likelihood Estimates

Variable	DF	Parameter Estimate	Standard Error	Chi-Square	Pr > ChiSq
treat	1	-1.79176	0.62361	8.2553	0.0041

Analysis of Maximum Likelihood Estimates

Variable	Hazard Ratio	95% Hazard Ratio Confidence Limits
treat	0.167	0.049 0.566

SAS program: lifetest vs phreg


```
options ls = 80;
libname fu '.';
data w1;
    set fu.aml;
proc lifetest notable;
    time survt*censor(0);
    strata treat;
proc phreg;
    model survt*censor(0) = treat/rl;
    strata pair;
run;
```

7. Example 11.2 (gastric cancer): Piece-wise Cox model

- (a) The data: survival time (days) from randomization to death from gastric cancer, censoring status, and treatment (0 - chemotherapy alone vs. 1 - chemotherapy and radiotherapy) (see Table 11.1).

- (b) Violation of PH assumption: see Figures 11.5 (KM plot) and 11.6 (log-cumulative hazard plot). P-value of testing PH assumption for treatment (from `cox.zph()`) is 0.0003.

Figure 1: Kaplan-Meier estimation of OS by treatment (Figure 11.5)



- (c) Piece-wise Cox model: Four pieces (0 - 360, 361 - 720, 721 - 1080, and 1080 - days); SAS program:

```
options ls = 78;  
libname fu '../..sdata';  
data fu.gastric;
```



```

infile '../..data//gastric.dat' LRECL = 30 misover pad;
input pid 1-3 os 8-12 censor 15 treat 20;

/* generating Figures 11.5 and 11.6 */
proc lifetest plot=(s, lls);
    time os * censor(0);
    strata treat;

/* cox.zph() to test PH assumption for treatment */
ex111.s<-function(){
    tmpdf<- importData("../..sdata/gastric.sas7bdat")
    fcox <- coxph(Surv(os, censor)~treat, data = tmpdf, x=T)
    zph <- cox.zph(fcox)
    wmf.graph("zph#.wmf")
    plot(zph)
    list(fcox, zph)
}

```

The outputs from Splus function ex111.s above are as follows:

```

[[1]]:
coxph(formula = Surv(os, censor) ~ treat, data = tmpdf, x = TRUE)
      coef exp(coef) se(coef)      z      p
treat 0.105      1.11    0.223 0.471 0.64

```

```

[[2]]:
      rho chisq      p
treat -0.397 13.1 0.000296

```

```

proc phreg;
    model os*censor(0) = x1 x2 x3 x4/risklimits;
    if 0 < os <= 360 then x1 = treat; else x1 = 0;
    if 360 < os <= 720 then x2 = treat; else x2 = 0;
    if 720 < os <= 1080 then x3 = treat; else x3 = 0;
    if os > 1080 then x4 = treat; else x4 = 0;
run;

```

SAS output:

Analysis of Maximum Likelihood Estimates

Parameter	DF	Parameter	Standard	Chi-Square	Pr > ChiSq
		Estimate	Error		
x1	1	0.87741	0.33515	6.8538	0.0088
x2	1	-0.25139	0.41684	0.3637	0.5464
x3	1	-1.10055	0.80231	1.8816	0.1701
x4	1	-1.18220	0.71109	2.7640	0.0964

Analysis of Maximum Likelihood Estimates

Parameter	Hazard	95% Hazard Ratio	
	Ratio	Confidence Limits	
x1	2.405	1.247	4.638
x2	0.778	0.344	1.760
x3	0.333	0.069	1.603
x4	0.307	0.076	1.236

Figure 2: Log-cumulative hazard by treatment (Figure 11.6)

